[bookmark: _Toc124048848]CONVERTIR UN NOMBRE EN LETTRES  ** (base Parrainage)
Option Compare Database
Option Explicit
Enum ARG_DEVISE
    Eurodevise = 0
    FrancDevise = 1
End Enum
Function ConvertNbLettres(Nb, Devise As ARG_DEVISE) As String
'Ecrit un nombre en toutes lettres à utiliser comme suit : ConvertNbLettres (22312.32, Eurodevise)
' gère également les nombres négatifs' limité à un nombre entre - 999 999 999 et 999 999 999
Dim stde As String
Dim stEspace As String
Dim stDevise As String
Dim stCentieme As String
Dim varnum, varnumD, varnumU, resultat, varlet
stEspace = " "
If Nb > 999999999 Or Nb < -999999999 Then
    MsgBox "Montant trop important !", vbInformation
    ConvertNbLettres = "Erreur"
    Exit Function
End If
Select Case Devise
    Case 0 'Euros
        stde = " d'"
        stDevise = "euro"
        stCentieme = "cent"
    Case 1 'Francs
        stde = " de"
        stDevise = "franc"
        stCentieme = "centime"
End Select
Static Chiffre(1 To 19)
    Chiffre(1) = "un"
    Chiffre(2) = "deux"
    Chiffre(3) = "trois"
    Chiffre(4) = "quatre"
    Chiffre(5) = "cinq"
    Chiffre(6) = "six"
    Chiffre(7) = "sept"
    Chiffre(8) = "huit"
    Chiffre(9) = "neuf"
    Chiffre(10) = "dix"
    Chiffre(11) = "onze"
    Chiffre(12) = "douze"
    Chiffre(13) = "treize"
    Chiffre(14) = "quatorze"
    Chiffre(15) = "quinze"
    Chiffre(16) = "seize"
    Chiffre(17) = "dix-sept"
    Chiffre(18) = "dix-huit"
    Chiffre(19) = "dix-neuf"
Static dizaine(1 To 8)
    dizaine(1) = "dix"
    dizaine(2) = "vingt"
    dizaine(3) = "trente"
    dizaine(4) = "quarante"
    dizaine(5) = "cinquante"
    dizaine(6) = "soixante"
    dizaine(8) = "quatre-vingt"
If Nb >= 1 Then   'traitement du cas 0 virgule quelque-chose
    resultat = ""
Else
    If Nb < 0 Then
        resultat = "moins"
        Nb = -Nb
    Else
        resultat = "zéro"
        GoTo FinTraitement
    End If
End If
varnum = Int(Nb / 1000000) 'traitement des millions
If varnum > 0 Then
    GoSub centaine_dizaine
    resultat = resultat + " " + varlet + " million"
    If varlet <> "un" Then: resultat = resultat + "s"
End If
varnum = Int(Nb) Mod 1000000  'traitement des milliers
varnum = Int(varnum / 1000)
If varnum > 0 Then
    GoSub centaine_dizaine
    If varlet <> "un" Then: resultat = resultat + " " + varlet
    resultat = resultat + " mille"
End If
varnum = Int(Nb) Mod 1000 'traitement des centaines et dizaines
If varnum > 0 Then
    GoSub centaine_dizaine
    resultat = resultat + " " + varlet
End If
resultat = LTrim(resultat)
varlet = Right$(resultat, 4)
Select Case varlet  'traitement du "s" final pour vingt et cent et du "d'" pour million
    Case "cent", "ingt"
        If Nb <> 20 And Nb <> 100 Then
            resultat = resultat + "s"
        End If
    Case "lion", "ions"
        resultat = resultat + stde
        stEspace = ""
End Select
FinTraitement:
resultat = resultat + stEspace + stDevise
If Nb > 2 Then: resultat = resultat + "s"
'traitement les centièmes
varnum = Int((Nb - Int(Nb)) * 100 + 0.5)
If varnum > 0 Then
    GoSub centaine_dizaine
    resultat = resultat + " et " + varlet + " " + stCentieme
    If varnum > 1 Then: resultat = resultat + "s"
End If
resultat = UCase(Left(resultat, 1)) + Right(resultat, Len(resultat) - 1) 'conversion 1er lettre en majuscule
ConvertNbLettres = resultat   'renvoi du résultat de la fonction et fin de la fonction
Exit Function
centaine_dizaine:   'sous programme
   varlet = ""
    If varnum >= 100 Then       'traitement des centaines
        varlet = Chiffre(Int(varnum / 100))
        varnum = varnum Mod 100
        If varlet = "un" Then
            varlet = "cent "
        Else
            varlet = varlet + " cent "
        End If
    End If
    If varnum <= 19 Then      'traitement des dizaines
        If varnum > 0 Then: varlet = varlet + Chiffre(varnum)
    Else
        varnumD = Int(varnum / 10)
        varnumU = varnum Mod 10
        Select Case varnumD
        Case Is <= 5
        varlet = varlet + dizaine(varnumD)
             Case 6, 7
                 varlet = varlet + dizaine(6)
             Case 8, 9
                 varlet = varlet + dizaine(8)
        End Select
        If varnumU = 1 And varnumD < 8 Then
             varlet = varlet + " et "
        Else
             If varnumU <> 0 Or varnumD = 7 Or varnumD = 9 Then: varlet =
varlet + " "
        End If
        If varnumD = 7 Or varnumD = 9 Then: varnumU = varnumU + 10
        If varnumU <> 0 Then: varlet = varlet + Chiffre(varnumU)
    End If
    varlet = RTrim(varlet)
    Return
End Function	

